UR1985_04

1985/04. GRIDCON: a FORTRAN program for converting one set of grid co-ordinates to another

J.W. Hudspeth

Abstract

GRIDCON is a Fortran program for converting survey points XY co-ordinates from one grid (*e.g.* local exploration grid) to another (*e.g.* AMG).

INTRODUCTION

The AMG (Australian Map Grid) is a rectangular co-ordinate system oriented N-S/E-W and is the accepted standard cartographic grid throughout Australia. Mapped data from geophysical surveys will usually be reduced and presented using AMG. However AMG is often not the ideal grid for data collection and local grids are set up specifically for this purpose. If the data collected on a local grid are to be reduced or presented using the AMG, then co-ordinate conversion will be required. The program GRIDCON was written to perform a conversion from one rectangular co-ordinate system to another.

CO-ORDINATE CONVERSION

Two rectangular grid co-ordinate systems will generally have differing origins and will be rotated with respect to one another. There may also be a scale difference between them although this should be very slight.

Consider two rectangular co-ordinate systems, XOY (being the system locating the survey points) and X'O'Y' (being the system we wish to use to represent the survey point positions). If the origin of XOY is at x'_o, y'_o of X'O'Y then transpose XOY so that the origins coincide. Let α be the angle between the positive y-axes of the co-ordinate systems where α is the rotation required to superimpose OY upon OY' and is negative if the sense of rotation is anti-clockwise and positive if it is clockwise.



Figure 1.

5 cm

From Figure 1 we see that

$$x' = x \cos \alpha - y \sin \alpha$$
and that $$y' = x \sin \alpha + y \cos \alpha$$

If one XOY unit of measurement is not quite equal to one X'O'Y unit of measurement

but $$(X'O'Y)_{\text{one unit}} = \text{FACTOR} \cdot (XOY)_{\text{one unit}}$$

then
$$x' = \text{FACTOR } (x \cos \alpha - y \sin \alpha)$$
$$y' = \text{FACTOR } (x \sin \alpha + y \cos \alpha)$$

Reversing the effect of the origin transposition, we have

$$x' = x_o' + \text{FACTOR } (x \cos \alpha - y \sin \alpha)$$
$$y' = y_o' + \text{FACTOR } (x \sin \alpha - y \cos \alpha)$$

RUNNING THE PROGRAM

The program is written in Fortran 77 and runs interactively on the Tasmania Department of Mines Perkin-Elmer 8/32 computer. The program is initiated by typing "GRIDCON" which starts a .CSS of the same name. The user then supplies the program with the necessary information in response to the screen prompts. After each pair of grid co-ordinates for conversion have been entered the user will be prompted to enter "N" (no) or "Y" (yes). "N" terminates the program and "Y" allows another pair of co-ordinates to be converted. The output on the line printer tabulates the co-ordinate pairs before and after conversion.

[23 January 1985]

APPENDIX 1

Program GRIDCON

```
C *
C PROGRAM GRIDCON
C CONVERTS THE POINTS OF ONE ORTHOGONAL GRID TO ANOTHER
C
      DOUBLE PRECISION OFFE,OFFN,TEAST,TNORT,SCALF,ANG,PI,CEAST,CNORT
     1,DEG,AMIN,SEC
      CHARACTER*1 ANS
      PI=3.14159D00
      WRITE(5,50)
   50 FORMAT(1X,'SPECIFY WHETHER ROTATION ANGLE IS RADIANS OR',
     1' DEGREES . . . .(R/D)')
   54 READ(5,55) ANS
   55 FORMAT(A1)
      IF (ANS.EQ.'R') THEN
      WRITE(5,58)
   58 FORMAT(1X,'ENTER ROTATION ANGLE IN RADIANS')
      READ(5,*) ANG
      WRITE(6,59) ANG
   59 FORMAT(1X,'ANGLE IN RADIANS  ',F9.4)
      ELSE
      IF(ANS.EQ.'D') THEN
      WRITE(5,60)
   60 FORMAT(1X,'ENTER ROTATION ANGLE IN DEGREES ,MINUTES,SECONDS',
     1,/,' IF ROTATION ANGLE IS NEGATIVE, ENTER DEGREES AS NEGATIVE',
     2' BUT MINUTES',/,' AND SECONDS AS POSITIVE')
      READ(5,*) DEG,AMIN,SEC
      WRITE(6,61) DEG,AMIN,SEC
   61 FORMAT(1X,'ANGLE IN DEGREES,MINUTES,SECONDS ',3F8.0)
      IF(NINT(DEG).LT.0) THEN
      AMIN=-AMIN
      SEC=-SEC
      ENDIF
      ANG=(PI/180.0D00)*(DEG+(AMIN+SEC/60.0D00)/60.0D00)
      ELSE
      WRITE(5,63)
   63 FORMAT(1X,'ANSWER R OR D')
      GO TO 54
      ENDIF
      ENDIF
      WRITE(5,64)
   64 FORMAT(1X,'ENTER GRIDS SCALING FACTOR')
      READ(5,*) SCALF
      WRITE(6,66) SCALF
   66 FORMAT(1X,'SCALE FACTOR ',F10.6)
      WRITE(5,67)
   67 FORMAT(1X,'ENTER EASTING OFFSET, NORTHING OFFSET')
      READ(5,*) OFFE,OFFN
      WRITE(6,68) OFFE,OFFN
   68 FORMAT(1X,'ORIGIN OFFSET EASTING ',F12.2,/,
     1' ORIGIN OFFSET NORTHING ',F12.2)
      WRITE(6,70)
   70 FORMAT(5X,'OLD EASTING    OLD NORTHING   NEW EASTING    ',
     1'NEW NORTHING')
   72 WRITE(5,73)
   73 FORMAT(1X,'ENTER GRID COORDINATES, EASTING THEN NORTHING')
      READ(5,*) CEAST,CNORT
      TEAST=SCALF*(CEAST*DCOS(ANG)-CNORT*DSIN(ANG))+OFFE
      TNORT=SCALF*(CEAST*DSIN(ANG)+CNORT*DCOS(ANG))+OFFN
      WRITE(6,120) CEAST,CNORT,TEAST,TNORT
  120 FORMAT(1X,4F15.0)
      WRITE(5,125)
  125 FORMAT(1X,'AGAIN . . . . . ?   (N/Y)')
  129 READ(5,55) ANS
      IF(ANS.EQ.'Y') GO TO 72
      IF(ANS.NE.'N') THEN
      WRITE(5,136)
  136 FORMAT(1X,'ANSWER N OR Y')
      GO TO 129
      ENDIF
      WRITE(6,200)
  200 FORMAT(1X)
      END
```